МНОГОПОТОЧНОСТЬ

Для тех, кто не слишком точно представляет, что такое многопоточность и как она работает, давайте сначала выясним, с чем же нам предстоит иметь дело. Компьютер с одним процессором может выполнять только одну задачу в единицу времени, однако ОС Windows представляет это так, как будто одновременно выполняются несколько задач. Это достигается посредством технологии так называемого временнóго квантования. Процессор выполняет одну задачу в течение определенного промежутка времени, затем переключается на другую задачу, которую выполняет в течение того же периода. Так продолжается до тех пор, пока число задач не исчерпано. По окончании списка одновременно выполняемых задач процессор возвращается к первой задаче и цикл повторяется.

Выполнение приложения CA-Visual Objects «с точки зрения процессора» является лишь одной из задач. Однако, внутри самого приложения может быть выполнена лишь одна задача за один промежуток времени. Многопоточность является способом разделения времени работы приложения на отдельные промежутки, в течение которых может выполняться более одной задачи. С многопоточностью связано множество потенциальных проблем, однако, если упростить решение задачи, то вы сможете избежать их.

Проблемой многопоточности с точки зрения ее визуального представления (в том числе и глав в книге) является то, что она не предназначена для демонстрации на экране монитора. Она изначально предназначена для фоновой работы, например, одновременного воспроизведения файла .WAV, мониторинга портов, направления задания печати на принтер и т.п. Однако, для демонстрации возможностей многопоточности нам необходимо использовать либо графический интерфейс GUI, либо терминальное окно, причем каждый из вариантов потенциально опасен. Свидетельством тому может быть приводимый ниже пример (открытие двух окон в двух различных потоках). Он работает на машинах с ОС Windows95 (200 MHz/128 Mb) и Windows NT (350 MHz/160Mb), однако с удручающим постоянством приводит к краху на машине с ОС Windows98 (550 MHz/192 Mb). Ко времени выпуска настоящей книги не удалось установить, с чем связана эта проблема – с операционной системой, процессором или чем-либо еще. (Прим. пер.: на моей машине то же самое – при завершении работы приложения появилось сообщение о выполнении недопустимой операции. Окно закрыть не удалось без перезагрузки машины. Win98 SE, Pentium-II – 300/128).

Имея в виду все сказанное выше, давайте приступим к практической реализации многопоточности в нашем собственном приложении.

В конце 21-й главы мы показали, как можно воспроизвести файл .WAV, однако проблемой являлось то, что нужно было дождаться окончания воспроизведения, прежде чем можно было сделать что-либо еще. С учетом продолжительности воспроизведения – 16 сек., такое положение не может быть признано удовлетворительным. Давайте посмотрим, сможет ли многопоточность помочь нам в решении этой проблемы.

ПРИМЕР СТАНДАРТНОГО МНОГОПОТОЧНОГО ПРИЛОЖЕНИЯ MDI
Находясь в репозитории, щелкните правой кнопкой мыши по проекту EVO, из контекстного меню выберите New Application, перейдите на вкладку Samples, выберите приложение Multithreaded Stadard MDI и нажмите кнопку ОК. Разверните в репозитории приложение MTStdApp, выделите модуль Standard Shell и откройте его метод FileOpen() в редакторе кода. Код должен выглядеть следующим образом:


METHOD FileOpen() CLASS StandardShellWindow

LOCAL dwThreadID AS DWORD
CreateVOThread(NULL_PTR, 0, @NewThreadWindow(), PTR(_CAST, SELF), 0, @dwThreadId)
Таким способом создается поток. Обратите внимание на то, что третий и шестой параметры передаются по ссылке, поэтому они префиксируются оператором @.

Третий параметр представляет собой имя функции, которая будет выполняться в новом потоке. Закройте редактор, откройте функцию NewThreadWindow() и посмотрите, что же происходит в этой функции. С ее помощью открывается стандартное диалоговое окно OpenDialog, в котором пользователь может выбрать файл .DBF, передать его в StdDataWindow, открыть окно данных DataWindow, после чего начинается выполнение петли Exec() для данного потока.

Это очень легко использовать, однако необходимо иметь в виду две важные особенности. Первая – это способ объявления функции:

STATIC FUNCTION NewThreadWindow(dwParam AS DWORD) AS INT PASCAL
Сильная типизация и соглашение о вызове PASCAL являются важными составными частями любой функции, передаваемой в функцию С. Вторая – это способ вызова метода Exec(). Любые ссылки на Exec() теперь должны оформляться именно таким образом, поскольку теперь у нас имеются несколько петель Exec(), выполняемых одновременно, и необходимо обеспечить обращение к нужному методу. Закройте редактор, скомпилируйте и запустите приложение.

Используя пункт Open в меню File, откройте два разных окна (на рис. 23.1 открыты два файла видео-библиотеки из C:\EVO\Video) и обратите внимание на заголовки окон. В них вы увидите идентификаторы различных потоков. 

Вы можете свободно переключаться между окнами, выполнять в них стандартные  операции, однако каждое из них работает в своей петле Exec(). По окончании тестирования (надеемся, ваш компьютер перенес его) закройте приложение и двинемся дальше.

ВИДЕТЬ ОДНО, СЛЫШАТЬ ДРУГОЕ

Приведенная выше демонстрация многопоточности продемонстрировала одновременную работу двух потоков, однако практическая ценность этого примера невелика. В нашем следующем примере мы постараемся достичь большей практичности. Возвращаясь к приложению Formula 1 (см. главу 21), напомним, что в конце главы мы немного развлеклись, воспроизводя звуковые файлы. Возвратитесь в модуль C:Start приложения Ch21 и откройте метод App:Start() в редакторе кода. Там вы обнаружите строку:

sndPlaySound(String2Psz("C:\Evo\Start.Wav"), 0)
Проблема заключалась в том, что во время воспроизведения 16-секундного файла ничего нельзя было сделать. Мы попытаемся решить эту проблему, воспроизводя файл в отдельном потоке с тем, чтобы остальная часть приложения могла бы работать самостоятельно. Измените эту строку так:

CreateVOThread(NULL_PTR, 0, @PlaySoundThread(), PTR(_CAST, SELF), 0, @dwThreadId)
Затем перейдите в начало метода и добавьте в список переменных еще одну:

LOCAL dwThreadID AS DWORD
Сохраните код, закройте редактор, щелкните правой кнопкой по модулю C:Start, выберите редактор кода из контекстного меню и введите следующий код:

STATIC FUNCTION PlaySoundThread(dwParam AS DWORD) AS INT PASCAL

sndPlaySound(String2Psz("C:\EVO\F1\Start.Wav"), 0)

RETURN
Сохраните код, закройте редактор, скомпилируйте  и запустите приложение.

Доказательством того, что все сработало, служит то, что мы одновременно слышим файл .WAV и видим окно приложения на экране. В процессе воспроизведения мы можем выполнять работу в приложении. Поток, в котором происходит воспроизведение файла, завершит свою работу по окончании воспроизведения, поскольку для этого потока не существует отдельного метода Exec(). Это работает точно также, как работало бы однопоточное приложение при отсутствии в нем метода App:Exec().

Это прекрасный способ заставить два процесса выполняться одновременно и в следующем примере попробуем развить его.

ОБРАБОТКА ФАЙЛОВ В ОТДЕЛЬНОМ ПРОЦЕССЕ

Создайте новый подкаталог “Ch23” в каталоге C:\EVO и скопируйте в него все файлы из каталога Ch23 с диска. Создайте новое приложение, назовите его “Ch23”, отключите все флажки и нажмите кнопку ОК.

Создайте в приложении новый модуль, назовите его “S:Emp1” и откройте редактор DB-серверов. Используя файлы Emp1.dbf и Emp1.ntx, создайте сервер, сохраните его и закройте редактор. Создайте другой модуль, назовите его “S:Emp2”, откройте редактор DB-серверов. Используя файлы Emp2.dbf и Emp2.ntx, создайте сервер, сохраните его и закройте редактор.

Во избежание появления предупреждений во время компиляции откройте методы Init() каждого из серверов и закомментируйте строки с объявлением локальной переменной lIndexFile в каждом из них. Сохраните изменения и закройте редактор. (Прим. пер:. мне не удалось обнаружить такую переменную ни в примере, ни в других приложениях. Компиляция прошла без предупреждений).
Создайте новый модуль, назовите его “W:Win1”, откройте оконный редактор, выберите тип окна DATAWINDOW и измените его имя на “Win1”. Нажмите кнопку ОК. Убедитесь, что размеры окна равны 420х320.

Разместите субокно в точек с координатами 10,10 на главном окне, задайте его размеры равными 400х120 и измените его имя на “sfOne”. В точке с координатами 10,140 разместите второе субокно и измените его имя на “sfTwo”. Растяните его до тех же размеров – 400х120. Сохраните окно и дважды щелкните по первому субнокну для запуска второго экземпляра оконного редактора.

Используя методику автоматической раскладки применительно к серверу Emp1, используйте только три первых его поля (EmpID, First_Name, Lаst_Name). Переведите окно в табличный режим просмотра и растяните колонки на всю ширину окна. Сохраните сделанные изменения, закройте редактор и возвратитесь к окну Win1. Теперь сделайте то же самое со вторым субкном, используя сервер Emp2. 

После завершения всех операций закройте все редакторы, перейдите в модуль Start, откройте метод App:Start() в редакторе кода и измените код следующим образом:

METHOD Start() CLASS App

LOCAL oMainWindow AS StandardShellWindow

LOCAL oWin AS Win1

oMainWindow := StandardShellWindow{SELF}

oMainWindow:Show(SHOWCENTERED)

oWin := Win1{oMainWindow}

oWin:Show()

SELF:Exec()

RETURN NIL
Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

На рис. 23.2 вы видите, что верхнее окно заполнено данными  из БД, в то время как второе – нет. По окончании тестирования закройте приложение.

Первая часть плана состоит в том, что при двойном щелчке мышью по какой-либо записи в верхнем окне она должна появляться в нижнем. Для реализации этого перейдите в модуль W:Win1, откройте редактор кода и введите следующее:

CLASS sfOneBrowser INHERIT DataBrowser

METHOD CellDoubleClick() CLASS sfOneBrowser

IF SELF:Owner:Owner:acsfTwo:Server:Append()

SELF:Owner:Owner:acsfTwo:Server:Emp_Id := SELF:Owner:Server:Emp_Id

SELF:Owner:Owner:acsfTwo:Server:First_Name := SELF:Owner:Server:First_Name

SELF:Owner:Owner:acsfTwo:Server:Last_Name := SELF:Owner:Server:Last_Name

SELF:Owner:Owner:acsfTwo:Server:Commit()

ENDIF
RETURN NIL

ACCESS acsfTwo( ) CLASS Win1

RETURN SELF:oSFsfTwo
Сохраните код, закройте редактор и нажмите кнопку Build. Посмотрев на код, вы можете обратить внимание на то, что требуется специальный метод ACCESS, поскольку мы находимся в методе броузера первого субокна. Необходимо создать новый класс броузера.

Откройте окно sfOne в редакторе и перейдите на вкладку DataWindow в окне его свойств. В свойстве Browser Inherits From укажите sfOneBrowser, сделав тем самым доступным наш метод CellDoubleClick. Сохраните окно, закройте редактор, скомпилируйте и запустите приложение.

При появлении на экране окна дважды щелкните по первой «реальной» записи (Richard Jones) и эта запись должна появиться в нижнем окне. На сей момент довольно. Закройте приложение и давайте доработаем его.

Вторая часть плана состоит в том, чтобы во время работы приложения можно было создать текстовый файл в отдельном потоке, не прерывая работы основного приложения. Для этого перейдите в модуль S:Emp2, откройте редактор кода и введите следующий код:

METHOD Append(lLocks) CLASS Emp2

LOCAL dwThreadID AS DWORD

LOCAL lRet

lRet := SUPER:Append(lLocks)

IF lRet

CreateVOThread(NULL_PTR, 0, @WriteFile(), PTR(_CAST, SELF), 0, @dwThreadId)

ENDIF

RETURN lRet
В этом методе мы перегружаем метод Append() сервера. Сначала он пытается добавить запись как обычно, и, в случае успеха, создает новый поток для выполнения функции с именем WriteFile(). Обратите внимание на следующий параметр – PTR(_CAST, SELF). Таким Образом можно передать вызывающий класс в качестве параметра. Отступите одну строку и введите следующее:

STATIC FUNCTION WriteFile( dwParam AS DWORD ) AS INT PASCAL

LOCAL oOwner AS Emp2

LOCAL ptrHandle AS PTR

LOCAL cFile AS STRING

oOwner := OBJECT(_CAST, dwParam)

Sleep(1000)

cFile := "C:\EVO\Ch23\F1" + oOwner:Emp_Id + ".txt"

IF ! (ptrHandle := FCreate(cFile)) = F_ERROR

FWrite(ptrHandle, "Hello " + oOwner:First_Name + CRLF )

FWrite(ptrHandle, "Your employee number is " + oOwner:Emp_Id + CRLF)

FClose(ptrHandle)

ENDIF

RETURN
Как видите, параметр, который ранее был преобразован в указатель, теперь вновь становится объектом.

Важно запомнить, что при использовании оператора _CAST только вы несете всю ответственность за то, чем все закончится. Другим важным замечанием является то, что эта функция объявлена как STATIC, соответственно, она должна находиться в том же модуле, что и класс, вызвавший ее. Если поместить ее в другой модуль, поведение приложения становится непредсказуемым.

Функция Sleep(1000) используется для создания небольшой паузы (1 сек.) для того, чтобы текст был выведен на экран прежде, чем новый поток попытается использовать его. Здесь же вы можете увидеть в работе низкоуровневые функции FCreate(), FWrite() и FClose(). Имейте в виду, что функция FCreate() либо создаст новый файл, либо перепишет уже существующий. Если вы хотите сохранить предыдущее содержимое файла, используйте функцию FOpen(). Сохраните код, закройте редактор, скомпилируйте и запустите приложение.


Прежде, чем продолжить, откройте Проводник Windows и перейдите в нем в каталог C:\EVO\Ch23 с тем, чтобы воочию наблюдать добавление файлов в каталог. Возвратитесь в приложение и дважды щелкните по одной из записей в верхнем окне.


Если все идет по плану, то вы должны увидеть, что в нижнем окне появится запись, а в каталоге C:\EVO\Ch23 добавится новый текстовый файл (рис. 23.4).

Закройте приложение и дважды щелкните по новому файлу, чтобы открыть есть с помощью ассоциированного приложения. Вы увидите текст, написанный с помощью функции FWrite() (рис. 23.5).

Закройте текстовый файл. и давайте разберем, чего же мы достигли и какие выводы можно сделать из этого.

На сей момент мы имеем событие, которое после непродолжительной паузы обеспечивает запись нового файла для каждой новой записи, появляющейся во втором сервере. В данном случае это приемлемо, однако вполне вероятна ситуация, при которой две записи будут добавлены друг за другом через очень непродолжительный промежуток времени. Это вызовет создание двух новых потоков, однако, если первый поток к этому времени будет уже готов к работе, то сервер к этому времени окажется на второй записи. Это приведет к тому, что вторая запись будет обработана дважды, а первая – ни разу. В следующем упражнении мы постараемся избавиться от этого недостатка.

Создайте новый модуль, назовите его “W:Win2”, откройте оконный редактор, выберите тип окна DATDWINDOW, измените его имя на “Win2” и нажмите ОК. Используйте сервер Emp2 и автораскладку, также как и ранее, используйте только три первых поля. Установите размеры окна равными 420х160, переключите его в табличный режим, растяните колонки до полной ширины окна и сохраните сделанное. Перейдите на вкладку  DataWindow в окне его свойств, откройте метод PostInit и измените код:

METHOD PostInit(oWindow,iCtlID,oServer,uExtra) CLASS Win2

//Put your PostInit additions here

SetTimer(SELF:Handle(), 6001, 3000, 0)

RETURN NIL
С помощью этого кода при открытии окна будет запущен таймер. Сохраните код, закройте редактор, перейдите на вкладку Window Events, откройте метод Close и введите следующее:

METHOD Close(oEvent) CLASS Win2

KillTimer(SELF:Handle(), 6001)

SUPER:Close(oEvent)

//Put your changes here

RETURN NIL
Обратите внимание, что таймер уничтожается прежде, чем следует обращение к SUPER. Сохраните код и закройте редактор. Закройте оконный редактор и откройте редактор кода.

Введите следующее:

METHOD Dispatch(oEvent) CLASS Win2

IF oEvent:Message == WM_TIMER

KillTimer(SELF:Handle(), 6001)

SELF:DoFiles()

SetTimer(SELF:Handle(), 6001, 3000, 0)

ENDIF

RETURN SUPER:Dispatch(oEvent)
Этот диспетчер перехватит событие таймера и немедленно уничтожит последний. Это сделано для предотвращения накопления в стеке событий таймера. Попытки обработки событий таймеров в то время, когда другое такое же событие уже обрабатывается – обычная причина краха приложений. После уничтожения таймера приложение вызывает метод DoFiles() (см. ниже), перезапускает таймер, а уже после этого следует обращение к SUPER.

Отступите одну строку вниз и введите код:

METHOD DoFiles() CLASS Win2

LOCAL ptrHandle AS PTR

LOCAL cFile AS STRING

SELF:Server:GoTop()

DO WHILE ! SELF:Server:Eof

cFile := "C:\EVO\Ch23\F1" + SELF:Server:Emp_Id + ".TXT"

IF ! (ptrHandle := FCreate(cFile)) = F_ERROR 

FWrite(ptrHandle, "Hello "+ SELF:Server:First_Name + CRLF)

FWrite(ptrHandle, "Your employee number is "+SELF:Server:Emp_Id+ CRLF)

FClose(ptrHandle)

ENDIF

SELF:Server:Skip()

ENDDO

RETURN NIL
В этом воде все знакомо, необходимо лишь обратить внимание, что сначала следует переход к первой записи сервера, а затем последователь обрабатываются все имеющиеся записи. Сохраните код, закройте редактор, перейдите в модуль Start, откройте метод App:Start() в редакторе и добавьте следующую строку непосредственно после объявления переменных:

SetExclusive(FALSE)
Сохраните код и закройте редактор. Теперь нам нужно отключить метод Append(), который мы использовали в предыдущем примере. Перейдите в модуль S:Emp2, откройте метод Append() и измените его имя на “wAppend (lLocks)”. Сохраните код и закройте редактор.

Теперь нужно обеспечить связь с новым окном. Перейдите в модуль Standard Shell, откройте объявление класса StandardShellWindow и измените его следующим образом:

CLASS StandardShellWindow INHERIT ShellWindow

PROTECT oPrinter AS PrintingDevice

PROTECT oWin2 AS Win2 
// Новая строка
Сохраните код, закройте редактор, откройте метод StandardShellWindow:Init(), перейдите в конец метод и измените строки следующим образом:

oPrinter := PrintingDevice{} 
// Существующая строка
SELF:oWin2 := Win2{SELF} 
// Новая строка
SELF:oWin2:Show() 

// Новая строка
RETURN SELF 


// Существующая строка
Этим способом обеспечивается открытие нового окна приложения при запуске окна оболочки. Отступите строку и введите строки:

METHOD Close(oEvent) CLASS StandardShellWindow

SELF:oWin2:EndWindow()

SUPER:Close(oEvent)

RETURN NIL
Это обеспечит закрытие окна приложения одновременно с закрытием окна оболочки. Сохраните код, закройте редактор, скомпилируйте и запустите приложение. Однако, вы можете обратить внимание на то, что теперь каждые три секунды текстовый файл перезаписывается. Это не совсем то, что планировалось. По окончании тестирования закройте приложение. При этом вы можете получить сообщение об ошибке, которое полностью зависит от таймера.

Проблема кроется в том, что метод EndWindow() посылает сообщение PostMessage, которое помещается на дно стека. Нам же нужно, чтобы оно помещалось на вершину стека. Откройте метод Close() и измените строку с EndWindow так:

SELF:oWin2:EndWindow(TRUE)
Это преобразует сообщение в SendMessage, которое помещается на вершину стека и, следовательно, обрабатывается первым. Сохраните код, закройте редактор и это должно решить проблему.

Другой частью проблемы является то, что записи обрабатываются повторно. Во избежание этого мы будем удалять записи непосредственно после их обработки. Это повлечет за собой большое количество удаленных записей, соответственно, необходим механизм повторного использования удаленных записей. Напомним, что при удалении записи она фактически не удаляется, а лишь помечается к удалению. Функция SetDeleted() позволяет таким записям быть видимыми или невидимыми. По умолчанию SetDeleted устанавливается в FALSE, т.е. удаленные записи видны.

Перейдите в модуль Start, откройте метод App:Start в редактор кода и добавьте следующую строку непосредственно после объявления переменных:

SetDeleted(TRUE)
Это исключит удаленные записи из области видимости. Сохраните код, закройте редактор, перейдите в модуль S:Emp2, откройте метод wAppend() и измените его так:

METHOD Append(lLocks) CLASS Emp2

LOCAL lRet AS LOGIC

SetDeleted(FALSE)

SELF:GoTop()

IF SELF:Deleted

lRet := SELF:Recall()

ELSE

lRet := SUPER:Append()

ENDIF

SetDeleted(TRUE)

RETURN lRet
Предполагая, что индекс определяет, что удаленные записи помещаются в начало индексного файла (или что ключевые поля обнуляются), мы видим, что вместо простого добавления записи код сначала проверяет наличие удаленных записей. Если таковые находятся, то нужное количество восстанавливается. Если же нет, то происходит обычное добавление записи.

Сохраните код и закройте редактор. Перейдите в модуль W:Win2, откройте метод DoFiles() и измените код:

METHOD DoFiles() CLASS Win2

LOCAL ptrHandle AS PTR

LOCAL cFile AS STRING

SELF:Server:GoTop()

DO WHILE ! SELF:Server:Eof

cFile := "C:\EVO\Ch23\F1" + SELF:Server:Emp_Id + ".txt"

IF ! (ptrHandle := FCreate(cFile)) = F_ERROR

FWrite(ptrHandle, "Hello " + SELF:Server:First_Name + CRLF )

FWrite(ptrHandle, "Your employee number is " + SELF:Server:Emp_Id + CRLF )

FClose(ptrHandle)

ENDIF

SELF:Server:Delete() 
// Новая строка
SELF:Server:Skip()

ENDDO

RETURN NIL
Сохраните код, закройте редактор, скомпилируйте и запустите приложение. Теперь вы обнаружите, что добавление и удаление записей может даже не успевать отображаться на экране. Однако, если вы проверите каталог, то увидите, что текстовые файлы исправно добавляются.

Закройте приложение и закончим на этом нашу главу.

ЗАКЛЮЧЕНИЕ

В этой главе вы увидели использование отдельных потоков для одновременного выполнения различных операций. Процессы, которые мы создали в этой главе, не слишком значительны, однако они могут быль расширены без особых проблем. 

Мы создали и запустили стандартное многопоточное приложение, хотя и предупреждали о нежелательности использования многопоточности для организации экранного вывода.

Звуковой файл, использованный в приложении Formula 1 был воспроизведен в отдельном потоке, что позволило запустить приложение во время исполнения музыки.

Последний пример показал, как обеспечить запись текстового файла в фоновом режиме, сначала -  с использованием отдельного потока, а затем – с использованием таймера.

Мы также увидели, как можно повторно использовать удаленные записи так, чтобы файл стал «самоподдерживающимся».

Мы еще раз обращаем ваше внимание на то, что многопоточность является комплексным процессом, который мы затронули лишь в незначительной степени и который следует использовать с большой осторожностью. Тем не менее, если стараться максимально упрощать код, как это сделали мы, вы избежите множества проблем.

